On considère un aménagement composé :

- d'un réservoir avec une vanne de 2 mètre de hauteur laissant passer un débit $q = 10$ m^2/s en O ;
- d'un coursier en pente raide ($i_1 = 5$ %) et moyennement rugueux (coefficient de Chézy $C = 50$ m$^{1/2}$.s^{-1}), d'une longueur de 10 m entre O et A ;
- d'un canal de pente douce ($i_1 = 0{,}2$ %) et de même rugosité rugueux que le coursier $C = 50$ m$^{1/2}$.s^{-1}, d'une longueur de 1000 m entre A et B ;
- d'un seuil d'une pelle $p = 0{,}5$ m en B.

Le coursier et le canal sont très larges.



Figure 1 : *aménagement étudié (échelle de longueur non respectée).*

On souhaite calculer la courbe de remous et notamment la position et les caractéristiques du ressaut. Pour cela on calcule les caractéristiques de l'écoulement :

- pour le coursier, on est en régime supercritique (torrentiel) : $h_n = 0{,}92$ m, $\mathrm{Fr}_0 = 1{,}12$, $\mathrm{Fr}_n = 3{,}6$;
- pour le canal, on est en régime subcritique (fluvial) : $h_n = 2{,}71$ m, $\mathrm{Fr}_n = 0{,}71$.

Pour l'ensemble de l'aménagement, la hauteur critique est la même et vaut :

$$h_c = \sqrt[3]{\frac{q^2}{g}} = 2{,}17 \text{ [m]},$$

Connaissant la hauteur d'écoulement à l'amont du coursier ($h = 2$ m), on peut calculer la courbe de remous en résolvant l'équation (1) numériquement :

$$\boxed{\frac{\mathrm{d}h}{\mathrm{d}x} = i\frac{1-(h_n/h)^3}{1-(h_c/h)^3},} \tag{1}$$

On trouve qu'en A, la hauteur vaut $h_A = 1{,}54$ m. On peut ensuite commencer à intégrer l'équation (1) pour le canal. Sans surprise, on trouve qu'il y a une transition critique au point C. On trouve numériquement $x_C = 90$ m. Pour calculer la position du ressaut, on commence par calculer l'autre branche reliant le point C à l'exutoire B. Au niveau du seuil le débit est « contrôlé » par la hauteur de p :

$$q = \sqrt{g}\left(\frac{2}{3}(H-p)\right)^{3/2} \quad [\mathrm{m}^2/\mathrm{s}],$$

ce qui implique que la charge totale H doit s'adapter à l'amont du seuil pour laisser transiter le débit q. On trouve qu'au voisinage de B, la charge H doit valoir $H = 3{,}73$ m, d'où l'on déduit que la hauteur avant le seuil doit être de $h_B = 3{,}25$ m. On calcule alors la courbe de remous entre A et B en résolvant l'équation (1) avec la condition à l'aval $h = h_B$ en B.

La position du front est trouvée en recherchant l'intersection de la courbe conjuguée (tracée en tireté sur la figure) de la courbe de remous AC avec la courbe de remous émanant de D. On trouve que l'intersection se fait en D' de coordonnée : $x_D = 24$ m. On relie les deux courbes de remous émanant de A et celle venant de B en considérant qu'elle se rejoignent au point D et qu'en ce point elles subissent un saut représenté par le segment DD' sur la figure 2. □



Figure 2 : *courbes de remous : solution donnée par l'équation (1) (courbe continue), courbe conjuguée (trait discontinue), et position du ressaut (courbe en gras).*

1. On commence par calculer les caractéristiques hydrauliques dans les deux biefs.

exemple.nb 1

```
In[19]:= q = 10;
         Ch = 50;
         i1 = 0.05;
         hn1 = (q / Ch / Sqrt[i1]) ^ (2 / 3)
         Frn = q / hn1^1.5 / Sqrt[9.81]
         hc = (q^2 / 9.81) ^ (1 / 3)
         Fr1 = q / 2^1.5 / Sqrt[9.81]

Out[22]= 0.928318

Out[23]= 3.56961

Out[24]= 2.16825

Out[25]= 1.12881

In[26]:= i2 = 0.002;
         hn2 = (q / Ch / Sqrt[i2]) ^ (2 / 3)
         Fr2 = q / hn2^1.5 / Sqrt[9.81]

Out[27]= 2.71442

Out[28]= 0.713922
```

2. On calcule la ligne d'eau dans le bief OA. On note que la hauteur en A vaut 1,54 m, donc elle est supérieure à la hauteur normale, mais inférieure à la hauteur critique, ce qui veut dire qu'en A l'écoulement est toujours supercritique.

exemple.nb 1

```
In[14]:= eqn1 = NDSolve[
           {h'[x] == i1 (1 - (hn1 / h[x]) ^3) / (1 - (hc / h[x]) ^3), h[0] == 2}, h[x], {x, 0, 100}]
         des0 = Plot[Evaluate[h[x] /. eqn1], {x, 0, 10}];
         hs = Evaluate[h[x] /. eqn1][[1]] /. x → 10

Out[14]= {{h[x] → InterpolatingFunction[{{0., 100.}}, <>][x]}}
```




```
Out[16]= 1.53911
```

3. On calcule la ligne d'eau dans le bief AB. Au point C, la routine de calcul s'arrête car une singularité est détectée (dénominateur tendant vers l'infini dans l'équation 1).

exemple.nb 1

```
In[20]:= eqn2 = NDSolve[
           {h'[x] == i2 (1 - (hn2 / h[x]) ^3) / (1 - (hc / h[x]) ^3), h[10] == hs}, h, {x, 10, 600}]
         xl = Flatten[h /. eqn2 /.
                HoldPattern[InterpolatingFunction[x__, y___]] -> x][[2]]
         des1 = Plot[Evaluate[h[x] /. eqn2], {x, 10, xl}, PlotRange -> {0, 3}];

        NDSolve::ndsz :
         At x == 90.30048673927307`, step size is effectively zero; singularity or stiff system suspected. Plus…

Out[20]= {{h -> InterpolatingFunction[{{10., 90.3005}}, <>]}}

Out[21]= 90.3005
```



4. On calcule la courbe conjuguée de la ligne d'eau dans le bief AB.

exemple.nb 1

```
In[26]:= conj[h_] := 1 / 2 * h * (Sqrt[8 * (q / h^1.5 / Sqrt[9.81]) ^2 + 1] - 1)
         des2 = Plot[conj@(Evaluate[h[x] /. eqn2][[1]]),
            {x, 10, xl}, PlotRange -> All, PlotStyle -> Dashing[{0.01, 0.01}]];
         Show[des0, des1, des2];
```





5. On calcule les caractéristiques hydrauliques au niveau du seuil.

```
In[48]:= p = 0.5;
         g = 9.81;
         Hf = (q)^(2/3)*3/2/g^(1/3) + p // N
         sol = h /. Solve[h + (q/h)^2/2/g == Hf, h]
         q/sol[[3]]^1.5/Sqrt[g]

Out[50]= 3.75238

Out[51]= {-1.03212, 1.50644, 3.27807}

Out[52]= 0.537945
```

6. On calcule la courbe de remous dans le bief AB.


```
In[70]:= eqn3 = NDSolve[{h'[x] == i2 (1 - (hn2/h[x])^3)/(1 - (hc/h[x])^3), h[1000] == sol[[3]]},
           h, {x, 1000, 10}]
         xl2 = Flatten[h /. eqn3 /.
                 HoldPattern[InterpolatingFunction[x__, y___]] → x][[1]]
         des3 = Plot[Evaluate[h[x] /. eqn3], {x, 1000, xl2}, PlotRange → All];
         des4 = Plot[conj@(Evaluate[h[x] /. eqn3][[1]]),
            {x, 1000, xl2}, PlotRange → {0, 3}, PlotStyle → Dashing[{0.01, 0.01}]];

Out[70]= {{h → InterpolatingFunction[{{10., 1000.}}, <>]}}

Out[71]= 10.
```





7. On peut tracer les courbes de remous et leur conjuguée. On note la symétrie de la représentation graphique.

```
In[57]:= des = Show[des0, des1, des2, des3, des4, Frame → True, Axes → False, FrameLabel →
            {StyleForm["  x  ", FontSize → 18, FontSlant -> "Italic", FontFamily → "Times",
              PrivateFontOptions → {"OperatorSubstitution" → False}],
             StyleForm[" h ", FontSize → 18, FontFamily → "Times", FontSlant -> "Italic",
              PrivateFontOptions → {"OperatorSubstitution" → False}]},
          DefaultFont → {"Times", 14}, ImageSize → 500];
```



8. On calcule le point d'intersection entre la courbe de remous (l'une des deux) et la conjuguée de l'autre courbe.


```
In[58]:= xr = x /. FindRoot[
            conj@(Evaluate[h[x] /. eqn3][[1]]) == Evaluate[h[x] /. eqn2], {x, 10, 90}][[1]]
         FindRoot[conj@(Evaluate[h[x] /. eqn2][[1]]) == Evaluate[h[x] /. eqn3], {x, 10, 90}]

Out[58]= 37.8227

Out[59]= {x → 37.8227}
```